
⭐⭐⭐ Défi ⏱️ 30 minutes

Capteur de parking luxe

La version luxe du capteur de parking ! Il bipe plus vite quand tu t’approches. Et l’écran OLED

montre la distance en centimètres. Avec une petite barre qui devient plus longue quand tu

t’approches.

🧰 De quoi as-tu besoin ?

1× Arduino UNO 1× Capteur de distance

HC-SR04

1× Buzzer actif

1× Petit écran OLED

0,91"

🔌
Câble USB + Câbles

jumper

🗺️ Schéma de branchement

🛠️ Voici comment le construire

1. Branche le HC-SR04 : Trig vers la broche 9, Echo vers la broche 10, Vcc

vers 5V et Gnd vers GND.

2. Branche le buzzer : S vers la broche 8, + vers 5V et − vers GND.

3. Branche l’écran OLED : SDA vers A4, SCL vers A5, VCC vers 5V et GND

vers GND.

4. Trois pièces ont besoin de 5V et de GND. Utilise les bandes + et − de ta

breadboard.

💻 Le code

🧠 Que fait le code ?

mesurerDistance()  mesure la distance en centimètres avec le HC-SR04.

afficherDistance()  met le nombre en grand sur l’écran et dessine une

petite barre avec fillRect() .

Le buzzer bipe plus vite quand la distance devient plus petite.

🎯 Essaie un peu…

Affiche "STOP!" sur l’écran si tu es à moins de 5 cm

Affiche la distance en mètres si quelque chose est à plus de 1 mètre

Ajoute l’anneau LED qui passe du vert au rouge

🚀 Construis ce projet avec Codey : ouvre codey.online/fr/starter-kits/uno/projects/parking-sensor-deluxe et clique sur FAIS ÇA. Alors le code et le

schéma de branchement sont tout de suite prêts.

Arduino UNO

Capteur de distance HC-SR04 Buzzer actif

Petit écran OLED 0,91"

// Capteur de parking luxe

// Comme le capteur de parking normal, mais maintenant tu vois aussi la 

distance sur l’écran OLED.

#include <Wire.h>

#include <Adafruit_GFX.h>

#include <Adafruit_SSD1306.h>

const int TRIG_PIN = 9;     // Trig du HC-SR04 sur la broche 9

const int ECHO_PIN = 10;    // Echo du HC-SR04 sur la broche 10

const int BUZZER_PIN = 8;   // S du buzzer sur la broche 8

Adafruit_SSD1306 ecran(128, 32, &Wire, -1);  // OLED: SDA = A4, SCL = A5

// Mesure la distance en centimètres

long mesurerDistance() {

  digitalWrite(TRIG_PIN, LOW);

  delayMicroseconds(2);

  digitalWrite(TRIG_PIN, HIGH);

  delayMicroseconds(10);

  digitalWrite(TRIG_PIN, LOW);

  long temps = pulseIn(ECHO_PIN, HIGH, 30000);

  if (temps == 0) return 999;   // Pas d’écho : rien tout près

  return temps / 58;

}

// Met la distance en grand sur l’écran, avec une petite barre dessous

void afficherDistance(long distance) {

  ecran.clearDisplay();

  ecran.setTextSize(2);

  ecran.setCursor(0, 0);

  if (distance >= 999) {

    ecran.print(F("---"));

  } else {

    ecran.print(distance);

    ecran.print(F(" cm"));

  }

  // Petite barre : plus c’est près, plus elle est longue

  int longueur = map(constrain(distance, 0, 100), 0, 100, 128, 0);

  ecran.fillRect(0, 24, longueur, 8, SSD1306_WHITE);

  ecran.display();

}

void setup() {

  pinMode(TRIG_PIN, OUTPUT);

  pinMode(ECHO_PIN, INPUT);

  pinMode(BUZZER_PIN, OUTPUT);

  ecran.begin(SSD1306_SWITCHCAPVCC, 0x3C);

  ecran.setTextColor(SSD1306_WHITE);

}

void loop() {

  long distance = mesurerDistance();

  afficherDistance(distance);

  if (distance < 5) {

    digitalWrite(BUZZER_PIN, HIGH);   // Très près : un long son

    delay(100);

  } else if (distance < 50) {

    digitalWrite(BUZZER_PIN, HIGH);   // Petit bip court

    delay(50);

    digitalWrite(BUZZER_PIN, LOW);

    delay(distance * 10);              // Plus près = bips plus vite

  } else {

    digitalWrite(BUZZER_PIN, LOW);

    delay(100);

  }

}

http://app/fr/starter-kits/uno/tutorial/kit/hc-sr04
http://app/fr/starter-kits/uno/tutorial/kit/active-buzzer-module
http://app/fr/starter-kits/uno/tutorial/kit/oled-display-091-inch

